
Zoo: A framework for the verification of concurrent
OCaml 5 programs using separation logic
Clément Allain #

INRIA Paris, France

Gabriel Scherer #

INRIA Paris, France

Abstract
The release of OCaml 5, which introduced parallelism into the language, drove the need for safe and
efficient concurrent data structures. New libraries like Saturn aim at addressing this need. From
the perspective of formal verification, this is an opportunity to apply and further state-of-the-art
techniques to provide stronger guarantees.

We present Zoo, a framework for verifying fine-grained concurrent OCaml 5 algorithms. We
followed a pragmatic approach, studying OCaml code written by concurrency expert to delimit a
limited but sufficient fragment of the language to express these algorithms: ZooLang. We formalized
its semantics carefully via a deep embedding in the Rocq proof assistant. We provide a tool to
translate source OCaml programs into ZooLang syntax inside Rocq, where they can be specified
and verified using the Iris concurrent separation logic.

We verified a subset of the standard library along with fine-grained concurrent algorithms,
including Treiber stack and a use of reference-counting for file descriptors from the Eio library. This
formalization work uncovered delicate questions of programming language semantics, especially
around physical equality. In the process, we also extended OCaml to more efficiently express certain
concurrent programs.

1 Introduction
Designing concurrent algorithms, in particular fine-grained concurrent algorithms, is a

notoriously difficult task. Similarly, the formal verification of such algorithms is also difficult.
It typically involves finding and reasoning about non-trivial linearization points [23, 32, 57,
58, 13].

In recent years, concurrent separation logic [7] has enabled significant progress in this
area. In particular, the development of Iris [31], a state-of-the-art mechanized higher-order
concurrent separation logic with user-defined ghost state, has nourished a rich and successful
line of works [32, 57, 58, 13, 8, 30, 51, 40, 39, 19, 45, 43, 42], dealing with external [58]
and future-dependent [32, 57, 13] linearization points, relaxed memory [40, 39, 19, 45] and
automation [43, 42].

Most of these works [32, 57, 58, 8, 30, 51, 43, 42] and many others [21, 47, 56, 37] rely
on HeapLang [54], the exemplar Iris language. HeapLang is a concurrent, imperative,
untyped, call-by-value functional language. To the best of our knowledge, it is currently the
closest language to OCaml 5 in the Iris ecosystem—we review the existing frameworks in
Section 2. It has been extended to handle weak memory [40] and algebraic effects [20].

Although HeapLang is theoretically expressive enough to represent OCaml programs,
our experiments showed that it is fairly impractical when it comes to verifying large OCaml
libraries. Indeed, it lacks basic abstractions such as algebraic data types (tuples, mutable and

immutable records, variants) and mutually recursive functions. Verifying OCaml programs
in HeapLang requires difficult translation choices and introduces various encodings, to the
point that the relation between the source and verified programs can become difficult to
maintain and reason about. It also has very few standard data structures that can be directly
reused. This view, we believe, is shared by many people in the Iris community. Our first
motivation in this work is therefore to fill this gap by providing a more practical OCaml-like
verification language: ZooLang. This language consists in a subset of OCaml 5 extended
with atomic record fields and equipped with a formal semantics and a program logic based
on Iris. We were influenced by the Perennial [10, 11, 12, 13] framework, which achieved
similar goals for the Go language with a focus on crash-safety. As in Perennial, we also
provide a translator from OCaml to ZooLang: ocaml2zoo. We call the resulting framework
Zoo.

Another, maybe less obvious, shortcoming of HeapLang is the soundness of its semantics
with respect to OCaml, in other words how faithful it is to the original language. One
ubiquitous—particularly in lock-free algorithms relying on low-level atomic primitives—and
subtle point is physical equality. In Section 5, we show that (1) HeapLang’s semantics for
physical equality is not compatible with OCaml and (2) OCaml’s informal semantics is
actually too imprecise to verify basic concurrent algorithms. To remedy this, we propose a
new formal semantics for physical equality and structural equality. We hope this work will
influence the way these notions are specified in OCaml.

In summary, we claim the following contributions:
1. We present ZooLang, a convenient subset of OCaml 5 formalized in Rocq (Sections 3

and 4). ZooLang comes with a program logic based on Iris and supports proof
automation through Diaframe [43, 42].

2. We provide a translator from OCaml to ZooLang: ocaml2zoo (Section 3), built for
practical applications—it supports full projects using the dune build system.

3. We formalize physical equality (Section 5) and structural equality (Section 6) in a faithful
way. To our knowledge this is the first detailed specification of physical equality for
a practical fragment of OCaml. The careful analysis of these notions suggests a new
OCaml feature: generative constructors.

4. We extend OCaml with atomic record fields and atomic arrays to ease the development
of fine-grained concurrent algorithms (Section 7).

5. We verify realistic use cases (Section 5) involving physical equality: (1) Treiber stack [9],
(2) a thread-safe wrapper around a file descriptor using reference-counting from the
Eio [38] library.

2 Related work
In general there are two approaches to practical program verification:

2.1 Non-automated verification
The verified program is translated, manually or in an automated way, into a representation

living inside a proof assistant. The user has to write specifications and prove them.
The representation may be primitive, like Gallina for Rocq. For pure programs, this

is rather straightforward, e.g. in hs-to-coq [52]. For imperative programs, this is more
challenging. One solution is to use a monad, e.g. in coq-of-ocaml [16], but it does not
support concurrency.

The representation may be embedded, meaning the semantics of the language is formalized

in the proof assistant. This is the path taken by some recent works [14, 26, 10, 18] harnessing
the power of separation logic. In particular, CFML [14] and Osiris [18] target OCaml.
However, CFML does not support concurrency and is not based on Iris. Osiris, still under
development, is based on Iris but does not support concurrency.

At the time of writing, HeapLang is thus the most appropriate tool to verify concurrent
OCaml programs. We discussed limitations of HeapLang in the introduction, and ZooLang
is our proposal to improve on this. Conversely, one notable limitation of ZooLang today is
its lack of support for OCaml’s relaxed memory model.

2.2 Semi-automated verification

In semi-automated verification approaches, the verified program is annotated by the
user to guide the verification tool: preconditions, postconditions, invariants, etc. Given
this input, the verification tool generates proof obligations that are mostly automatically
discharged. One may further distinguish two types of semi-automated systems: foundational
and non-foundational.

In non-foundational automated verification, the tool and the external solvers it may
rely on are part of the trusted computing base. It is the most common approach and has
been widely applied in the literature [53, 44, 28, 22, 3, 24, 36, 48], including to OCaml by
Cameleer [46], which uses the Gospel specification language [15] and Why3 [24].

In foundational automated verification, the proofs are checked by a proof assistant like
Rocq, meaning the automation does not have to be trusted. To our knowledge, it has been
applied to C [49] and Rust [25].

Zoo is a non-automated verification framework—except for our use Diaframe for local
automation of separation logic reasoning. We would be interested in moving towards more
automation in the future.

2.3 Physical equality

There is some literature in proof-assistant research on reflecting physical equality from
the implementation language into the proof assistant, for optimization purposes: for example,
exposing OCaml’s physical equality as a predicate in Rocq lets us implement some
memoization and sharing techniques in Rocq libraries. However, axiomatizing physical
equality in the proof assistant is difficult, and can result in inconsistencies.

The earlier discussions of this question that we know come from Jourdan’s thesis [29]
(chapter 9), also presented more succintly in [6]. This work introduces the Jourdan condition,
that physical equality implies equality of values. [5] extends the treatment of physical equality
in Rocq, integrating it in an “extraction monad” to control it more safely. There is also a
discussion of similar optimizations in Lean in [50].

The correctness of the axiomatization of physical equality depends on the type of the
values being compared: axiomatizations are typically polymorphic on any type A, but their
correctness depends on the specific A being considered. For example, it is easy to correctly
characterize physical on natural numbers, and other non-dependent types arising in Rocq
verification projects. One difficulty in HeapLang and ZooLang is that they are untyped
languages, their representation of 0 and false has the same type. But our remark that
structural equality (in OCaml) does not necessarily coincide with definitional equality (in
Rocq) also applies to other Rocq types: our examples with an existential Any constructor
(see Section 5) can be reproduced with Σ-types.

Rocq term t

constructor C

projection proj
record field fld
identifier s, f ∈ String
integer n ∈ Z
boolean b ∈ B
binder x ::= <> | s

unary operator ⊕ ::= ~ | -
binary operator ⊗ ::= + | - | * | ‘quot‘ | ‘rem‘ | ‘land‘ | ‘lor‘ | ‘lsl‘ | ‘lsr‘

| <= | < | >= | > | = | ̸= | == | !=
| and | or

expression e ::= t | s | #n | #b

| fun: x1 . . . xn => e | rec: f x1 . . . xn => e | e1 e2

| let: x := e1 in e2 | e1 ;; e2

| let: f x1 . . . xn := e1 in e2 | letrec: f x1 . . . xn := e1 in e2

| let: ‘C x1 . . . xn := e1 in e2 | let: x1, . . . ,xn := e1 in e2

| ⊕e | e1 ⊗ e2

| if: e0 then e1 (else e2)?

| for: x := e1 to e2 begin e3 end
| §C | ‘C (e1, . . . ,en) | (e1, . . . ,en) | e.<proj>
| [] | e1 :: e2

| ‘C {e1, . . . ,en} | {e1, . . . ,en} | e.{fld} | e1 <-{fld} e2

| ref e | !e | e1 <- e2

| match: e0 with br1| . . . | brn (|_ (as s)? => e)? end
| e.[fld] | Xchg e1 e2 | CAS e1 e2 e3 | FAA e1 e2

| Proph | Resolve e0 e1 e2

branch br ::= C (x1 . . . xn)? (as s)? => e

| [] (as s)? => e | x1 :: x2 (as s)? => e

toplevel value v ::= t | #n | #b

| fun: x1 . . . xn => e | rec: f x1 . . . xn => e

| §C | ‘C (v1, . . . ,vn) | (v1, . . . ,vn)
| [] | v1 :: v2

Figure 1 ZooLang syntax (omitting mutually recursive toplevel functions)

3 Zoo in practice
3.1 Language

The core of Zoo is ZooLang: a concurrent, imperative, untyped, functional programming
language fully formalized in Rocq. Its semantics has been designed to match OCaml’s.

ZooLang comes with a program logic based on Iris: reasoning rules expressed in
separation logic (including rules for the different constructs of the language) along with
Rocq tactics that integrate into the Iris proof mode [35, 34]. In addition, it supports
Diaframe [43, 42], enabling proof automation.

The ZooLang syntax is given in Figure 11, omitting mutually recursive toplevel functions

1 More precisely, it is the syntax of the surface language, including Rocq notations.

that are treated specifically. Expressions include standard constructs like booleans, integers,
anonymous functions (that may be recursive), applications, let bindings, sequence, unary
and binary operators, conditionals, for loops, tuples. In any expression, one can refer to a
Rocq term representing a ZooLang value (of type val) using its Rocq identifier. ZooLang
is deeply embedded: variables (bound by functions and let) are quoted as strings.

Data constructors (immutable memory blocks) are supported through two constructs : §C

represents a constant constructor (e.g. §None), ‘C (e1, . . . ,en) represents a non-constant
constructor (e.g. ‘Some( e )). Unlike OCaml, ZooLang has projections of the form
e.<proj> (e.g. (x, y).<1>), that can be used to obtain a specific component of a tuple or
data constructor. ZooLang supports shallow pattern matching (patterns cannot be nested)
on data constructors with an optional fallback case.

Mutable memory blocks are constructed using either the untagged record syntax {e1, . . . ,en}
or the tagged record syntax ‘C {e1, . . . ,en}. Reading a record field can be performed using
e.{fld} and writing to a record field using e1 <-{fld} e2. Pattern matching can also be used
on mutable tagged blocks provided that cases do not bind anything—in other words, only
the tag is examined, no memory access is performed. References are also supported through
the usual constructs : ref e creates a reference, !e reads a reference and e1 <- e2 writes
into a reference. The syntax seemingly does not include constructs for arrays but they are
supported through the Array standard module (e.g. array_make).

Note that ZooLang follows OCaml in sometimes eschewing orthogonality to provide
more compact memory representations: constructors are n-ary instead of taking a tuple as
parameter, and the tagged record syntax is distinct from a constructor taking a mutable record
as parameter. In each case the simplifying encoding would introduce an extra indirection in
memory, which is absent from the ZooLang semantics. Performance-conscious experts care
about these representation choices, and we care about faithfully modeling their programs.

Parallelism is mainly supported through the Domain standard module (e.g. domain_spawn),
including domain-local storage. Special constructs (Xchg, CAS, FAA; see Section 4.4) are used
to model atomic references.

The Proph and Resolve constructs model prophecy variables [32], see Section 4.5.

3.2 Translation from OCaml to ZooLang
While ZooLang lives in Rocq, we want to verify OCaml programs. To connect them

we provide the tool ocaml2zoo to translate OCaml source files2 into Rocq files containing
ZooLang code. This tool can process entire dune projects, and support several libraries
provided together or as dependencies of the project.

The supported OCaml fragment includes: tuples, variants, records and inline records,
shallow match, atomic record fields, unboxed types, toplevel mutually recursive functions.

Consider, for example, the OCaml implementation of a concurrent stack [9] in Figure 2.
The push function is translated into:
Definition stack_push : val := rec: "push" "t" "v" =>

let: "old" := !"t" in
let: "new_" := "v" :: "old" in
if: ~ CAS "t".[contents] "old" "new_" then (

domain_cpu_relax () ;;
"push" "t" "v" ).

2 Actually, ocaml2zoo processes binary annotation files (.cmt files).

type 'a t = 'a list Atomic.t
let create () = Atomic.make []

let rec push t v =
let old = Atomic.get t in
let new_ = v :: old in
if not (Atomic.compare_and_set t old new_) then (

Domain.cpu_relax () ;
push t v

)

let rec pop t =
match Atomic.get t with
| [] -> None
| v :: new_ as old ->

if Atomic.compare_and_set t old new_ then (
Some v

) else (
Domain.cpu_relax () ;
pop t

)

Figure 2 Implementation of a concurrent stack

3.3 Specifications and proofs
Once the translation to ZooLang is done, the user can write specifications and prove

them in Iris. For instance, the specification of the stack_push function could be:
Lemma stack_push_spec t ι v :

<<< stack_inv t ι

| ∀∀ vs, stack_model t vs >>>
stack_push t v @ ↑ι

<<< stack_model t (v :: vs)
| RET (); True >>>.

Proof. ... Qed.
Here, we use a logically atomic specification [17], which has been proven [4] to be equivalent

to linearizability [27] in sequentially consistent memory models.
Similarly to Hoare triples, the specification is formed of a precondition and a postcondition,

represented in angular brackets. But each is split in two parts, a public or atomic condition,
and a private condition. Following standard Iris notations, the private conditions are on
the outside (first line of the precondition, last line of the postcondition) and the atomic
conditions are inside.

For this particular operation, the private postcondition is trivial. The private condition
stack_inv t is the stack invariant. Intuitively, it asserts that t is a valid concurrent stack.
More precisely, it enforces a set of logical constraints—a concurrent protocol—that t must
respect at all times.

The atomic pre- and post-conditions specify the linearization point of the operation:
during the execution of stack_push, the abstract state of the stack held by stack_model is
atomically updated from vs to v :: vs: v is atomically pushed at the top of the stack.

https://iris-project.org/

4 Zoo features
In this section, we review the salient features of Zoo, which we found lacking when we

attempted to use HeapLang to verify real-world OCaml programs. We start with the most
generic ones and then address those related to concurrency.

4.1 Algebraic data types
Zoo is an untyped language but, to write interesting programs, it is convenient to work

with abstractions like algebraic data types. To simulate tuples, variants and records, we
designed a machinery to define projections, constructors and record fields.

For example, one may define a list-like type with:
Notation "'Nil'" := (in_type "t" 0) (in custom zoo_tag).
Notation "'Cons'" := (in_type "t" 1) (in custom zoo_tag).

Users do not need to write this incantation directly, as they are generated by ocaml2zoo
from the OCaml type declarations. Suffice it to say that it introduces the two tags in the
zoo_tag custom entry, on which the notations for data constructors rely. The in_type term
is needed to distinguish the tags of distinct data types; crucially, it cannot be simplified away
by Rocq, as this could lead to confusion during the reduction of expressions.

Given this incantation, one may directly use the tags Nil and Cons in data constructors
using the corresponding ZooLang constructs:
Definition map : val :=

rec: "map" "fn" "t" =>
match: "t" with
| Nil => §Nil
| Cons "x" "t" =>

let: "y" := "fn" "x" in
‘Cons( "y", "map" "fn" "t" )

end.
Similarly, one may define a record-like type with two mutable fields f1 and f2:

Notation "'f1'" := (in_type "t" 0) (in custom zoo_field).
Notation "'f2'" := (in_type "t" 1) (in custom zoo_field).

Definition swap : val :=
fun: "t" =>

let: "f1" := "t".{f1} in
"t" <-{f1} "t".{f2} ;; "t" <-{f2} "f1".

4.2 Mutually recursive functions
Zoo supports non-recursive (fun: x1 . . . xn => e) and recursive (rec: f x1 . . . xn => e)

functions but only toplevel mutually recursive functions. It is non-trivial to properly handle
mutual recursion: when applying a mutually recursive function, a naive approach would
replace calls to sibling functions by their respective bodies, but this typically makes the
resulting expression unreadable. To prevent it, the mutually recursive functions have to
know one another to preserve their names during β-reduction. We simulate this using some
boilerplate that can be generated by ocaml2zoo. For instance, one may define two mutually
recursive functions f and g as follows:
Definition f_g := (

recs: "f" "x" => "g" "x"

and: "g" "x" => "f" "x"
)%zoo_recs.

(* boilerplate *)
Definition f := ValRecs 0 f_g.
Definition g := ValRecs 1 f_g.
Instance : AsValRecs' f 0 f_g [f;g]. Proof. done. Qed.
Instance : AsValRecs' g 1 f_g [f;g]. Proof. done. Qed.

4.3 Standard library
To save users from reinventing the wheel, we provide a standard library—more or less a

subset of the OCaml standard library. Currently, it mainly includes standard data structures
like: array (Array), resizable array (Dynarray), list (List), stack (Stack), queue (Queue),
double-ended queue, mutex (Mutex), condition variable (Condition).

Each of these standard modules contains ZooLang functions and their verified specifications.
These specifications are modular: they can be used to verify more complex data structures.
As an evidence of this, lists [1] and arrays [2] have been successfully used in verification
efforts based on Zoo.

4.4 Concurrent primitives
Zoo supports concurrent primitives both on atomic references (from Atomic) and atomic

record fields (from Atomic.Loc3) according to the table below. The OCaml expressions
listed in the left-hand column translate into the Zoo expressions in the right-hand column.
Notice that an atomic location [%atomic.loc e.f] (of type _ Atomic.Loc.t) translates
directly into e.[f].

OCaml Zoo
Atomic.get e !e

Atomic.set e1 e2 e1 <- e2

Atomic.exchange e1 e2 Xchg e1.[contents] e2

Atomic.compare_and_set e1 e2 e3 CAS e1.[contents] e2 e3

Atomic.fetch_and_add e1 e2 FAA e1.[contents] e2

Atomic.Loc.exchange [%atomic.loc e1.f] e2 Xchg e1.[f] e2

Atomic.Loc.compare_and_set [%atomic.loc e1.f] e2 e3 CAS e1.[f] e2 e3

Atomic.Loc.fetch_and_add [%atomic.loc e1.f] e2 FAA e1.[f] e2

One important aspect of this translation is that atomic accesses (Atomic.get and
Atomic.set) correspond to plain loads and stores. This is because we are working in a
sequentially consistent memory model: there is no difference between atomic and non-atomic
memory locations.

4.5 Prophecy variables
Lock-free algorithms exhibit complex behaviors. To tackle them, Iris provides powerful

mechanisms such as prophecy variables [32]. Essentially, prophecy variables can be used to
predict the future of the program execution and reason about it. They are key to handle

3 The Atomic.Loc module is part of the PR that implements atomic record fields (see Section 7).

https://iris-project.org/
https://github.com/ocaml/ocaml/pull/13404


C. Allain and G. Scherer 23:9

future-dependent linearization points: linearization points that may or may not occur at a
given location in the code depending on a future observation.

Zoo supports prophecy variables through the Proph and Resolve expressions—as in
HeapLang, the canonical Iris language. In OCaml, these expressions correspond to
Zoo.proph and Zoo.resolve, that are recognized by ocaml2zoo.

5 Physical equality
The notion of physical equality is ubiquitous in fine-grained concurrent algorithms. It

appears not only in the semantics of the == operator, but also in the semantics of the
Atomic.compare_and_set primitive, which atomically sets an atomic reference to a desired
value if its current content is physically equal to an expected value. This primitive is
commonly used to try committing an atomic operation in a retry loop, as in the push and
pop functions of Figure 2.

5.1 Physical equality in HeapLang
In HeapLang, this primitive is provided but restricted. Indeed, its semantics is

only defined if either the expected or the desired value fits in a single memory word
in the HeapLang value representation: literals (booleans, integers and pointers4) and
literal injections5; otherwise, the program is stuck. In practice, this restriction forces the
programmer to introduce an indirection [55, 32, 57] to physically compare complex values,
e.g. lists. Furthermore, when the semantics is defined, values are compared using their Rocq
representations; physical equality boils down to Rocq equality.

5.2 Physical equality in OCaml
In OCaml, physical equality is more tricky and often considered dangerous. Structural

equality, which we describe in Section 6, should be the preferred way of comparing values.
However, structural equality is typically much slower than physical equality, as it basically
compiles to only one assembly instruction. Also, the Atomic.compare_and_set requires the
comparison to be atomic, which is the case for physical equality but not structural equality.

In particular, the semantics of physical equality is non-deterministic. To see why, consider
the case of immutable blocks representing constructors and immutable records (as opposed to
mutable blocks representing mutable records), e.g. Some 0. The physical comparison of two
seemingly identical immutable blocks, according to the Rocq representation (essentially a
tag and a list of fields), may return false. Indeed, at runtime, a non-empty immutable block
is represented by a pointer to a tagged memory block. In this case, physical equality is just
pointer comparison. It is clear that two pointers being distinct does not imply the pointed
memory blocks are. In other words, we cannot determine the result of physical comparison
just by looking at the abstract values.

The question is then: what guarantees do we get when physical equality returns true and
when it returns false? Given such guarantees, denoted by val_physeq and val_physneq,
the non-deterministic semantics is reflected in the logic through the following specification:
Lemma physeq_spec v1 v2 :

{{{ True }}}

4 HeapLang allows arbitrary pointer arithmetic and therefore inner pointers. This is forbidden in both
OCaml and ZooLang, as any reachable value has to be compatible with the garbage collector.

5 HeapLang has no primitive notion of constructor, only pairs and injections (left and right).

v1 == v2
{{{ b, RET #b; ⌜(if b then val_physeq else val_physneq) v1 v2⌝ }}}

Proof. ... Qed.
The OCaml manual documents a partial specification for physical equality, which is

precise for basic types such as references, but does not clearly extend to structured values
containing a mix of immutable and mutable constructors. The only guarantee that it provides
for all values is: if two values are physically equal, they are also structurally equal. This
means we don’t learn anything when two values are physically distinct.

In the following, we will explore both cases, looking at the optimizations that the compiler
or the runtime system may perform. We will show that the aforementioned guarantee is
arguably not sufficient to verify interesting concurrent programs and attempt to establish
stronger guarantees.

5.3 When physical equality returns true
Let us go back to the concurrent stack of Figure 2 and more specifically the push

function. To prove the atomic specification given in Section 3, we rely on the fact that,
if Atomic.compare_and_set returns true, we actually observe the same list of values in
the sense of Rocq equality. However, assuming only structural equality as per OCaml’s
specification of physical equality, this cannot be proven. To see why, consider, e.g., a stack
of references ('a ref). As structural equality is indeed structural, it traverses the references
without comparing their physical identities. In other words, we cannot conclude the references
are exactly the same. Hence, we cannot prove the specification.

This conclusion might seem surprising and counterintuitive. Indeed, we know that physical
equality essentially boils down to a comparison instruction, so we should be able to say
more. Departing from OCaml’s imprecise specification, let us attempt to establish stronger
guarantees. We assume the following classification of values: booleans, integers, mutable
blocks (pointers), immutable blocks, functions.

The easy cases are mutable blocks and functions. Each of these two classes is disjoint
from the others. We can reasonably assume that, when physical equality returns true and
one of the compared values belongs to either of these classes, the two values are actually the
same in Rocq. As far as we are aware, there is no optimization that could break this.

Booleans, integers and empty immutable blocks are represented by immediate integers
through an encoding. This encoding induces conflicts: two seemingly distinct values in Rocq
may have the same encoding. For example, the following tests all return true (Obj.repr is
an unsafe primitive revealing the memory representation of a value):
let test1 = Obj.repr false == Obj.repr 0 (* true *)
let test2 = Obj.repr None == Obj.repr 0 (* true *)
let test3 = Obj.repr [] == Obj.repr 0 (* true *)

The semantics of unrestricted physical equality has to reflect these conflicts. In our
experience, restricting compared values similarly to typing is quite burdensome; the specification
of polymorphic data structures using physical equality has to be systematically restricted. In
summary, when physical equality on immediate values returns true, it is guaranteed that
they have the same encoding.

Finally, let us consider the case of non-empty immutable blocks. At runtime, they are
represented by pointers to tagged memory blocks. At first approximation, it is tempting to
say that physically equal immutable blocks really are definitionally equal in Rocq. Alas, this
is not true. To explain why, we have to recall that the OCaml compiler and the runtime
system (e.g., through hash-consing) may perform sharing: immutable blocks containing

https://rocq-prover.org/
https://rocq-prover.org/
https://rocq-prover.org/

physically equal fields may be shared. For example, the following tests may return true:
let test1 = Some 0 == Some 0 (* true *)
let test2 = [0;1] == [0;1] (* true *)

On its own, sharing is not a problem. However, coupled with representation conflicts, it
can be surprising. Indeed, consider the any type defined as:
type any = Any : 'a -> any

The following tests may return true:
let test1 = Any false == Any 0 (* true *)
let test2 = Any None == Any 0 (* true *)
let test3 = Any [] == Any 0 (* true *)

Now, going back to the push function of Figure 2, we have a problem. Given a stack of
any, it is possible for the Atomic.compare_and_set to observe a current list (e.g., [Any 0])
physically equal to the expected list (e.g., [Any false]) while these are actually distinct in
Rocq. In short, the expected specification of Section 3 is incorrect. To fix it, we would need
to reason modulo physically equality, which is non-standard and quite burdensome.

We believe this really is a shortcoming, at least from the verification perspective. Therefore,
we propose to extend OCaml with generative immutable blocks6. These generative blocks
are just like regular immutable blocks, except they cannot be shared. Hence, if physical
equality on two generative blocks returns true, these blocks are definitionally equal in Rocq.
At user level, this notion is materialized by generative constructors. For instance, to verify
the expected push specification, we can use a generative version of lists:
type 'a list =

| Nil
| Cons of 'a * 'a list [@generative]

5.4 When physical equality returns false
Most formalizations of physical equality in the literature do not give any guarantee when

physical equality returns false. Many use-cases of physical equality, in particular retry
loops, can be verified with only sufficient conditions on true. However, in some specific
cases, more information is needed.

Consider the Rcfd module from the Eio [38] library, an excerpt of which is given in
Figure 37. Thomas Leonard, its author, suggested that we verify this real-life example
because of its intricate logical state. However, we found out that it is also relevant regarding
the semantics of physical equality. Essentially, it consists in wrapping a file descriptor in
a thread-safe way using reference-counting. At creation in the make function, the wrapper
starts in the Open state. At some point, it can switch to the Closing state in the close
function and can never go back to the Open state. Crucially, the Open state does not change
throughout the lifetime of the data structure.

The interest of Rcfd lies in the close function. First, the function reads the state. If
this state is Closing, it returns false; the wrapper has been closed. If this state is Open, it
tries to switch to the Closing state using Atomic.Loc.compare_and_set; if this attempt
fails, it also returns false. In this particular case, we would like to prove that the wrapper
has been closed, or equivalently that Atomic.Loc.compare_and_set cannot have observed
Open. Intuitively, this is true because there is only one Open.

6 https://github.com/clef-men/ocaml/tree/generative_constructors
7 We make use of atomic record fields as introduced in Section 7.1.

type state =
| Open of Unix.file_descr
| Closing of (unit -> unit)

type t =
{ mutable ops: int [@atomic];

mutable state: state [@atomic]; }

let make fd = { ops = 0; state = Open fd }

let closed = Closing (fun () -> ())
let close t =

match t.state with
| Closing _ -> false
| Open fd as prev ->

let next = Closing (fun () -> Unix.close fd) in
if Atomic.Loc.compare_and_set [%atomic.loc t.state] prev next then (

if t.ops == 0
&& Atomic.Loc.compare_and_set [%atomic.loc t.state] next closed
then close () ;
true

) else false

Figure 3 Rcfd module from Eio [38] (excerpt)

Obviously, we need some kind of guarantee related to the physical identity of Open when
Atomic.Loc.compare_and_set returns false. If Open were a mutable block, we could argue
that this block cannot be physically distinct from itself; no optimization we know of would
allow that. Unfortunately, it is an immutable block, and immutable blocks are subject to
more optimizations. In fact, something surprising but allowed8 by OCaml can happen:
unsharing, the dual of sharing. Indeed, any immutable block can be unshared, that is
reallocated. For example, the following test may theoretically return false:
let x = Some 0
let test = x == x (* false *)

Going back to Rcfd, we have a problem: in the second branch, the Open block corresponding
to prev could be unshared, which would make Atomic.Loc.compare_and_set fail. Hence,
we cannot prove the expected specification; in fact, the program as it is written has a bug.

To remedy this unfortunate situation, we propose to reuse the notions of generative
immutable blocks, that we introduced to prevent sharing, to also forbid unsharing by the
OCaml compiler – we implemented this in an experiment branch of OCaml.

In our semantics, each generative block is annotated with a logical identifier9 representing
its physical identity, much like a pointer for a mutable block. If physical equality on two
generative blocks returns false, the two identifiers are necessarily distinct. Given this
semantics, we can verify the close function. Indeed, if Atomic.Loc.compare_and_set fails,
we now know that the identifiers of the two blocks, if any, are distinct. As there is only one

8 This has been confirmed by OCaml experts developing the Flambda backend.
9 Actually, for practical reasons, we distinguish identified and unidentified generative blocks.

Open block whose identifier does not change, it cannot be the case that the current state is
Open, hence it is Closing. We can verify this function after adding the following annotation:
type state =

| Open of Unix.file_descr [@generative]
| Closing of (unit -> unit)

5.5 Summary
In summary, we give the following specification to physical equality in ZooLang, which

also serves as a precise specification of physical equality of a practical fragment of OCaml:

On values whose low-level representation is an immediate integer, physical equality is
immediate equality.
On values whose low-level representation are mutable blocks at some location, or generative
immutable blocks with some identity, physical equality is equality of locations or identities.
On values whose low-level representation are immutable blocks, physical-equality is under-
specified, but it implies that the blocks that the same tags and that their arguments are
in turn physically equal.
Two values that fall into different categories above are never physically equal.

6 Structural equality
Structural equality is also supported. More precisely, it is not part of the semantics of

the language but axiomatized on top of it10. The reason is that it is in fact difficult to
specify for arbitrary values. In general, we have to compare graphs—which implies structural
comparison may diverge.

Accordingly, the specification of v1 = v2 requires the (partial) ownership of a memory
footprint corresponding to the union of the two compared graphs, giving the permission to
traverse them safely. If it terminates, the comparison decides whether the two graphs are
isomorphic (modulo representation conflicts, as described in Section 5). In Iris, this gives:
Axiom structeq_spec : ∀ v1 v2 footprint,

val_traversable footprint v1 →
val_traversable footprint v2 →
{{{ structeq_footprint footprint }}}

v1 = v2
{{{ b, RET #b;

structeq_footprint footprint ∗
⌜(if b then val_structeq else val_structneq) footprint v1 v2⌝ }}}.

Obviously, this general specification is not very convenient to work with. Fortunately,
for abstract values (without any mutable part), we can prove a much simpler variant saying
that structural equality boils down to physical equality:
Lemma structeq_spec_abstract v1 v2 :

val_abstract v1 →
val_abstract v2 →
{{{ True }}}

v1 = v2

10 We could also have implemented it in ZooLang, but that would require more low-level primitives.

{{{ b, RET #b; ⌜(if b then val_physeq else val_physneq) v1 v2⌝ }}}
Proof. ... Qed.

7 OCaml extensions for fine-grained concurrent programming

Over the course of this work, we studied efficient fine-grained concurrent OCaml programs
written by experts. This revealed various limitations of OCaml in these domains, that
those experts would work around using unsafe casts, often at the cost of both readability
and memory-safety; and also some mismatches between their mental model of the semantics
of OCaml and the mental model used by the OCaml compiler authors. We worked on
improving OCaml itself to reduce these work-arounds or semantic mismatches.

7.1 Atomic record fields

OCaml 5 offers a type 'a Atomic.t of atomic references exposing sequentially-consistent
atomic operations. Data races on non-atomic mutable locations has a much weaker semantics
and is generally considered a programming error. For example, the Michael-Scott concurrent
queue [41] relies on a linked list structure that could be defined as follows:

type 'a node = Nil | Cons of { value : 'a; next : 'a node Atomic.t }

Performance-minded concurrency experts dislike this representation, because 'a Atomic.t
introduces an indirection in memory: it is represented as a pointer to a block containing the
value of type 'a. Instead, they use something like the following:

type 'a node = Nil | Cons of { mutable next: 'a node; value: 'a }
let as_atomic : 'a node -> 'a node Atomic.t option = function

| Nil -> None
| (Next _) as record -> Some (Obj.magic record : 'a node Atomic.t)

Notice that the next field of the Cons constructor has been moved first in the type
declaration. Because the OCaml compiler respects field-declaration order in data layout, a
value Cons { next; value } has a similar low-level representation to a reference (atomic
or not) pointing at next, with an extra argument. The code uses Obj.magic to unsafely cast
this value to an atomic reference, which appears to work as intended.

Obj.magic is a shunned unsafe cast (the OCaml equivalent of unsafe or unsafePerformIO).
It is very difficult to be confident about its usage given that it may typically violate
assumptions made by the OCaml compiler and optimizer. In the example above, casting
a two-fields record into a one-argument atomic reference may or may not be sound—but
it gives measurable performance improvements on concurrent queue benchmarks. (TODO:
benchmark to quantify the improvement.)

It is possible to statically forbid passing Nil to as_atomic to avoid error handling,
by turning 'a node into a GADT indexed over it a type-level representation of its head
constructor. Examples of this pattern can be found in the Kcas [33] library by Vesa Karvonen.
It is difficult to write correctly and use, in particular as unsafe casts can sometimes hide
type-errors in the intended static discipline.

Note that this unsafe approach only works for the first field of a record, so it is not
applicable to records that hold several atomic fields, such as the toplevel record storing
atomic front and back pointers for the concurrent queue.

https://github.com/ocaml-multicore/kcas

7.1.1 Our atomic fields proposal

We proposed a design for atomic record fields as an OCaml language change proposal:
RFC #3911. Declaring a record field atomic simply requires an [@atomic] attribute—and
could eventually become a proper keyword of the language.
(* re-implementation of atomic references *)
type 'a atomic_ref = { mutable contents : 'a [@atomic]; }

(* concurrent linked list *)
type 'a node = Nil | Cons of { value: 'a; mutable next : 'a node [@atomic]; }

(* bounded SPSC circular buffer *)
type 'a bag =

{ data : 'a Atomic.t array;
mutable front: int [@atomic];
mutable back: int [@atomic]; }

The design difficulty is to express atomic operations on atomic record fields. For example,
if buf has type 'a bag above, then one naturally expects the existing notation buf.front to
perform an atomic read and buf.front <- n to perform an atomic write. But how would
one express exchange, compare-and-set and fetch-and-add? We would like to avoid adding a
new primitive language construct for each atomic operation.

Our proposed implementation12 introduces a built-in type 'a Atomic.Loc.t for an atomic
location that holds an element of type 'a, with a syntax extension [%atomic.loc <expr>.<field>]
to construct such locations. Atomic primitives operate on values of type 'a Atomic.Loc.t,
and they are exposed as functions of the module Atomic.Loc.

For example, the standard library exposes
val Atomic.Loc.fetch_and_add : int Atomic.Loc.t -> int -> int
and users can write:
let preincrement_front (buf : 'a bag) : int =

Atomic.Loc.fetch_and_add [%atomic.loc buf.front] 1

where [%atomic.loc buf.front] has type int Atomic.Loc.t. Internally, a value of type
'a Atomic.Loc.t can be represented as a pair of a record and an integer offset for the
desired field, and the atomic.loc construction builds this pair in a well-typed manner.
When a primitive of the Atomic.Loc module is applied to an atomic.loc expression, the
compiler can optimize away the construction of the pair—but it would happen if there was
an abstraction barrier between the construction and its use.

Note: the type 'a Atomic.t of atomic references exposes a function
val Atomic.make_contended : 'a -> 'a Atomic.t

that ensures that the returned atomic value is allocated with enough alignment and padding
to sit alone on its cache line, to avoid performance issues caused by false sharing. Currently
there is no such support for padding of atomic record fields (we are planning to work on this
if the support for atomic fields gets merged in standard OCaml), so the less-compact atomic
references remain preferable in certain scenarios.

11 https://github.com/ocaml/RFCs/pull/39
12 https://github.com/ocaml/ocaml/pull/13404

7.2 Atomic arrays
On top of our atomic record fields, we have implemented support for atomic arrays,

another facility commonly requested by authors of efficient concurrent programs. Our previous
example of a concurrent bag of type 'a bag used a backing array of type 'a Atomic.t array,
which contains more indirections than may be desirable, as each array element is a pointer
to a block containing the value of type 'a, instead of storing the value of type 'a directly in
the array.

Our implementation of atomic arrays13 builds on top of the type 'a Atomic.Loc.t we
described in the previous section, and it relies on two new low-level primitives provided by
the compiler:
val Atomic_array.index : 'a array -> int -> 'a Atomic.Loc.t
val Atomic_array.unsafe_index : 'a array -> int -> 'a Atomic.Loc.t

The function index takes an array and an integer index within the array, and returns an
atomic location into the corresponding element after performing a bound check. unsafe_index
omits the boundcheck—additional performance at the cost of memory-safety—and allows to
express the atomic counterpart of the unsafe operations Array.unsafe_get and Array.unsafe_set.
The atomic primitives of the module Atomic.Loc can then be used on these indices; our
implementation implements a library module on top of these primitives to provide a higher-
level layer to the user, with direct array operations such as:
val Atomic_array.exchange : 'a Atomic_array.t -> int -> 'a -> 'a
val Atomic_array.unsafe_exchange : 'a Atomic_array.t -> int -> 'a -> 'a

8 Conclusion and future work
We presented Zoo, a framework for the verification of concurrent OCaml 5 programs.

While it is not yet available on opam, it can be installed and used in other Rocq projects.
We provide a minimal example14 demonstrating its use.

Zoo has already been used to verify sequential imperative algorithms [1] and is currently
being used to verify a library of lock-free data structures. Its main weakness so far is its
memory model, which is sequentially consistent as opposed to the relaxed OCaml 5 memory
model. It also lacks exceptions and algebraic effects, that we plan to introduce in the future.

Another interesting direction would be to combine Zoo with semi-automated techniques.
Similarly to Why3, the simple parts of the verification effort would be done in a semi-
automated way, while the most difficult parts would be conducted in Rocq.

